

Unidade 4 - Algoritmos Resolvidos

Algoritmo "PositivoNegativo"

Var

Funcao positivoNegativo : caractere

Var

numero : inteiro

Inicio

escreva("Número:")

leia(numero)

se (numero < 0) entao

retorne "NEG"

senao

retorne "POS"

fimse

Fimfuncao

Inicio

escreva(positivoNegativo)

Fimalgoritmo

Algoritmo "SomaDoisEntre"

Var

numero1, numero2 : inteiro

Funcao somaEntre(n1, n2 : inteiro) : inteiro

Var

aux1, aux2, i, soma : inteiro

Inicio

aux1 <- n1 + 1

aux2 <- n2 - 1

soma <- 0

para i <- aux1 ate aux2 passo 1 faca

soma <- soma + i

fimpara

retorne soma

Fimfuncao

Inicio

escreva("Numero 1:")

leia(numero1)

escreva("Numero 2:")

leia(numero2)

escreva(somaEntre(numero1,numero2))

Fimalgoritmo

Algoritmo "MediaNotas"

Var

n1, n2, n3 : real

tipomedia : caractere

Funcao mediaNotas(nota1, nota2, nota3 : real; tipo : caractere) : real

Var

media : real

Inicio

media <- 0

se tipo = "A" entao

media <- (nota1 + nota2 + nota3) / 3

senao

se tipo = "P" entao

media <- (nota1 * 1 + nota2 * 3 + nota3 * 6) / 10

senao

media <- 0

fimse

fimse

retorne media

Fimfuncao

Inicio

escreva("Nota 1:")

leia(n1)

escreva("Nota 2:")

leia(n2)

escreva("Nota 3:")

leia(n3)

escreva("Tipo da média:")

leia(tipomedia)

escreva(mediaNotas(n1,n2,n3,tipomedia))

Fimalgoritmo

Algoritmo "PercentualAumento"

Var

antigo, atual : real

procedimento aumento(valorAntigo, valorAtual : real)

Var

percentual : real

Inicio

percentual <- valorAtual * 100 / valorAntigo - 100

escreva("Percentual de aumento: ", percentual)

Fimprocedimento

Inicio

escreva("Valor antigo:")

leia(antigo)

escreva("Valor atual:")

leia(atual)

aumento(antigo, atual)

Fimalgoritmo

Algoritmo "NumerosMaiorMenor"

Var

procedimento maiorMenor

Var

i, numero, maior, menor : inteiro

Inicio

para i <- 1 ate 5 passo 1 faca

 escreva("Informe o número:")

 leia(numero)

 se (i = 1) entao

 maior <- numero

 menor <- numero

 senao

 se (numero > maior) entao

 maior <- numero

 fimse

 se (numero < menor) entao

 menor <- numero

 fimse

 fimse

fimpara

 escreva("Maior número é ", maior, " e o menor é ", menor)

Fimprocedimento

Inicio

 maiorMenor

Fimalgoritmo

Algoritmo "VetorFatorial"

Var

A : vetor [1..10] de inteiro

B : vetor [1..10] de inteiro

i : inteiro

procedimento entradaVetorA

Var

i : inteiro

Inicio

para i <- 1 ate 10 passo 1 faca

 escreva("Numero: ")

 leia(A[i])

fimpara

Fimprocedimento

funcao fatorial (N : inteiro) : inteiro

var

i, fat : inteiro

inicio

fat <- 1

para i <- 1 ate N passo 1 faca

 fat <- fat * i

fimpara

retorne fat

fimfuncao

procedimento geraVetorB

Var

i : inteiro

Inicio

para i <- 1 ate 10 passo 1 faca

B[i] <- fatorial(A[i])

fimpara

fimprocedimento

Inicio

entradaVetorA

geraVetorB

escreva("Valores do vetor B")

para i <- 1 ate 10 passo 1 faca

escreva (B[i])

fimpara

Fimalgoritmo

algoritmo "Operacao"

var

numero1, numero2 : real

operacao : caractere

funcao calculo(n1, n2 : real; simbolo : caractere) : real

inicio

escolha(simbolo)

caso ("+")

retorne $n1 + n2$

caso ("-")

retorne $n1 - n2$

caso ("*")

retorne $n1 * n2$

caso ("/")

retorne $n1 / n2$

outrocaso

retorne -1

fimescolha

fimfuncao

inicio

escreva("Informe o primeiro número:")

leia (numero1)

escreva("Informe o segundo número:")

leia (numero2)

escreva("Informe o símbolo da operação:")

leia (operacao)

escreva (calculo(numero1,numero2,operacao))

fimalgoritmo

algoritmo "ComparaCadeia"

var

palavra1, palavra2 : caractere

funcao compare(palavra1, palavra2 : caractere) : inteiro

inicio

se (palavra1 = palavra2) entao

retorne 1

senao

retorne 0

fimse

fimfuncao

inicio

escreva("Informe a primeira palavra:")

leia(palavra1)

escreva("Informe a segunda palavra:")

leia(palavra2)

se (compare(palavra1,palavra2) = 1) entao

escreva("As palavras são iguais")

senao

escreva("As palavras são diferentes")

fimse

fimalgoritmo

algoritmo "PesoHomemMulher"

var

altura : real

sexo : caractere

funcao pesoIdeal(altura : real; sexo : caractere) : real

var

aux : real

inicio

se (sexo = "H") entao

aux <- 72.7 * altura - 58

senao

se (sexo = "M") entao

aux <- 62.1 * altura - 44.7

senao

aux <- -1

fimse

fimse

retorne aux

fimfuncao

inicio

escreva("Informe a altura:")

leia(altura)

escreva("Informe o sexo (H ou M):")

leia(sexo)

escreva ("Peso ideal: " , pesoIdeal(altura,sexo))

fimalgoritmo

algoritmo "pesquisa"

var

sexo : vetor[1..5] de caractere

corOlhos : vetor[1..5] de caractere

corCabelos: vetor[1..5] de caractere

idade: vetor[1..5] de inteiro

procedimento entrada()

var i : inteiro

inicio

para i <- 1 ate 5 passo 1 faca

 escreva("Sexo:")

 leia (sexo[i])

 escreva("Cor dos olhos:")

 leia (corOlhos[i])

 escreva("Cor dos cabelos:")

 leia (corCabelos[i])

 escreva("Idade:")

 leia (idade[i])

fimpara

fimprocedimento

funcao media() : real

var

i, soma , contador : inteiro

inicio

para i <- 1 ate 5 passo 1 faca

 se (corOlhos[i] = "C") E (corCabelos[i] = "P") entao

 soma <- soma + idade[i]

 contador <- contador + 1

fimse

fimpara

retorne soma / contador

fimfuncao

funcao maior() : inteiro

var

i, idadeaux : inteiro

inicio

para i <- 1 ate 5 passo 1 faca

se (i = 0) entao

idadeaux <- idade[i]

senao

Se (idade[i] > idadeaux) entao

idadeaux <- idade[i]

fimse

fimse

fimpara

retorne idadeaux

fimfuncao

funcao qtdFeminino() : inteiro

var

i, contador : inteiro

inicio

para i <- 1 ate 5 passo 1 faca

Se (sexo[i] = "Feminino") E (idade[i] >= 18) E (idade[i] <= 35) E

(corOlhos[i] = "A") E (corCabelos[i] = "L") entao

contador <- contador + 1

fimse

fimpara

retorne contador

fimfuncao

inicio

entrada()

escreva ("Média de idade das pessoas com olhos castanhos e cabelos pretos:", media())

escreva ("Maior idade entre os habitantes:", maior())

escreva ("Qtd.Mulheres com idade entre 18 e 35 anos com olhos azuis e cabelos loiros",
qtdFeminino())

fimalgoritmo

algoritmo "ordem"

var

numero : vetor[1..5] de inteiro

ind : inteiro

procedimento ordenacao()

var

x, i, j : inteiro

inicio

para i <- 1 ate 4 passo 1 faca

para j <- i + 1 ate 5 passo 1 faca

se (numero[j] < numero[i]) entao

x <- numero[i]

numero[i] <- numero[j]

numero[j] <- x

fimse

fimpara

fimpara

fimprocedimento

inicio

para ind <- 1 ate 5 passo 1 faca

escreva("Informe o número:")

leia (numero[ind])

fimpara

ordenacao()

para ind <- 1 ate 5 passo 1 faca

escreva(numero[ind])

fimpara

fimalgoritmo

```

algoritmo "valorreferencia"
var
    numero : inteiro
procedimento valor (numero : inteiro)
var
    i : inteiro
inicio
    para i <- 1 ate 3 passo 1 faca
        numero <- numero * i
    fimpara
fimprocedimento
procedimento referencia (var numero : inteiro)
var
    i : inteiro
inicio
    para i <- 1 ate 3 passo 1 faca
        numero <- numero * i
    fimpara
fimprocedimento
inicio
    escreva("Informe o número:")
    leia(numero)
    valor(numero)
    escreva("Número: ",numero)
    referencia(numero)
    escreva("Número: ",numero)
finalgoritmo

```

Na passagem de parâmetro por valor, o parâmetro formal recebe uma cópia do parâmetro real, sendo assim, qualquer alteração no parâmetro formal não afeta o parâmetro real. Por referência, o parâmetro formal recebe o endereço do parâmetro real, sendo assim, qualquer alteração no parâmetro formal reflete em alteração do parâmetro real.

Algoritmo "Somatorio"

Var

i : inteiro

funcao somatorio (v: inteiro): inteiro

inicio

se (v = 1) entao

retorne 1

senao

retorne v + somatorio(v-1)

fimse

fimfuncao

Inicio

escreva(somatorio(4))

Fimalgoritmo

Algoritmo "Potencia"

Var

i : inteiro

funcao potencia (base, expoente : inteiro): inteiro

inicio

se (expoente = 0) entao

retorne 1

senao

retorne base * potencia(base, expoente - 1)

fimse

fimfuncao

Inicio

escreva(potencia(3,4))

Fimalgoritmo

Algoritmo "Fibonacci"

Var

i : inteiro

funcao fibonacci (v: inteiro): inteiro

inicio

se (v = 0) ou (v = 1) entao

retorne 1

senao

retorne fibonacci(v-1) + fibonacci(v-2)

fimse

fimfuncao

Inicio

para i <- 0 ate 19 passo 1 faca

escreva(fibonacci(i), " ")

fimpara

Fimalgoritmo